

```

60080 ' Initialisierung *****
60090 MODE 1:INK 0,1:INK 1,24:INK 2,5,16:INK 3,21
60100 PAPER 0:PEN 3:BORDER 2:SPEED INK 30,20
60110 SYMBOL AFTER 253
60120 SYMBOL 253,&66,&0,&78,&C,&7C,&CC,&76,&0
60130 SYMBOL 254,&66,&0,&66,&66,&66,&66,&3E,&0
60140 SYMBOL 255,&66,&0,&3C,&66,&66,&66,&3C,&0
60150 LOCATE 1,1:PRINT STRINGS(40,210);
60160 LOCATE 1,2:PRINT CHR$(24)+SPACES(17)+"JUMPREF"+SPACES(16)+CHR$(24)
60170 LOCATE 10,5:PRINT CHR$(164);" Martin Kotulla 1985"
60180 PRINT STRINGS(40,210)
60190 PEN 2:PRINT " JUMPREF";:PEN 1:PRINT " erstellt von Ihren Basic-
60200 PRINT:PRINT " Programmen Sprunglisten mit allen"
60210 PRINT:PRINT " Querverweisen auf Zeilennummern."
60220 PRINT:PRINT " Das Programm mu"+CHR$(177)+" mit MERGE hinzu-"
60230 PRINT:PRINT " gef"+CHR$(254)+"gt werden und niedrigere Zeiten-"
60240 PRINT:PRINT " nummern als JUMPREF haben."
60250 PEN 3:PRINT STRINGS(40,210);
60260 PEN 3:PRINT:PRINT " Bitte w"+CHR$(253)+"hlen Sie:";
60270 PEN 1:PRINT " 1- Bildschirm":PRINT:PRINT TAB(19);" 2- Druckerausgabe";
60280 PEN 2:PRINT " ";CHR$(143)
60290 in$=UPPER$(INKEYS):IF in$="" THEN 60290
60300 IF in$="1" THEN dev=0:GOTO 60340
60310 IF in$<>"2" THEN PRINT CHR$(7);:GOTO 60290
60320 dev=8:LOCATE 1,20:PRINT CHR$(20):PRINT
60330 PEN 3:PRINT TAB(5);"BITTE WARTEN - JUMPREF ARBEITET!":GOTO 60360
60340 MODE 2:INK 1,0:INK 0,10:PEN 1:PAPER 0:BORDER 10
60350 ' Hauptprogramm: Zeilen analysieren *****
60360 start=368 ' Startadresse aller Basic-Programme
60370 length=PEEK(start)+PEEK(start+1)*256
60380 linenum=PEEK(start+2)+PEEK(start+3)*256
60390 IF linenum>59999 THEN 60610
60400 shown=0
60410 i=start+5:WHILE i<=start+length-1
60420 token=PEEK(i)
60430 IF token=&19 THEN i=i+1:GOTO 60560 ' ***** Token fuer Ein-Byte-Wert
60440 IF token=&FF THEN i=i+1:GOTO 60560 ' ***** Token fuer Funktionen
60450 IF token=&1A THEN i=i+2:GOTO 60560 ' ***** Dezimaler Zwei-Byte-Wert
60460 IF token=&1B THEN i=i+2:GOTO 60560 ' ***** Binaerer Zwei-Byte-Wert
60470 IF token=&1C THEN i=i+2:GOTO 60560 ' ***** Hexadezimaler Zwei-Byte-Wert
60480 IF token=&1F THEN i=i+5:GOTO 60560 ' ***** Fließkommazahl
60490 IF token<>&1E AND token<>&1D THEN 60510
60500 IF NOT shown THEN PRINT #dev,USING "##### ";linenum;:shown=-1
60510 IF token<>&1E THEN 60520
60515 PRINT #dev,USING "##### ";PEEK(i+1)+PEEK(i+2)*256;:i=i+1
60520 IF token<>&1D THEN 60560
60530 adr=PEEK(i+1)+PEEK(i+2)*256+3
60540 ln=PEEK(adr)+PEEK(adr+1)*256:PRINT #dev,USING "##### ";ln;
60550 i=i+1
60560 i=i+1:WEND
60570 start=start+length
60580 IF shown THEN PRINT #dev
60590 GOTO 60370
60600 ' Programmende *****
60610 PRINT CHR$(7):PRINT #dev:PRINT #dev,"*EOF*"
60620 END ' *****

```